Microsoft KB Archive/291106

= XL97: How to Use the EFFECT and NOMINAL Functions to Calculate Loan or Mortgage Interest with Intrayear Compounding =

Article ID: 291106

Article Last Modified on 9/25/2006

-

APPLIES TO


 * Microsoft Excel 97 Standard Edition

-



This article was previously published under Q291106



SUMMARY
The future value of a dollar amount, commonly called the compounded value, involves the application of compound interest to a present value amount. The result is a future dollar amount. Three types of compounding include: annual, intrayear, and annuity compounding. This article discusses intrayear calculations for compound interest.



Calculating Future Value with Intrayear Compounded Interest
Intrayear compound interest is interest that is compounded more often than once a year. Financial institutions may calculate interest based on semiannual, quarterly, monthly, weekly, or even daily compounding. For example, mortgage payments are calculated differently in different jurisdictions, particularly in the United States and Canada, with the result that effective interest rates are not the same for loans or mortgages that are described as having the same nominal rate.

Microsoft Excel includes the EFFECT function in the Analysis ToolPak add-in. The EFFECT function returns the effective compounded interest rate based on the nominal annual interest rate and the number of compounding periods per year. This is the same as the simple interest that would be paid on the loan as a lump sum.

The NOMINAL function performs the inverse calculation; it returns the nominal interest rate based on the effective rate and the number of compounding periods.

The general equation to calculate the future value of a loan or mortgage is as follows   =P*(1+(k/m))^(m*n) where

P = initial principal

k = nominal annual interest rate

m = number of times per year the interest is compounded

n = number of years or term of the loan

You can use the EFFECT worksheet function to obtain the same result as follows   =P+(P*EFFECT(EFFECT(k,m)*n,n)) or to determine the effective interest rate as follows:   =EFFECT(EFFECT(k,m)*n,n) When you deal with a long-term loan, use the EFFECT function twice: first to determine the effect of intrayear compounding within a particular year, and then to extend that effect over the term of the loan. For a one-year calculation, you can use the simpler form:   =EFFECT(k,m)

Examples
To calculate future value and effective interest by using the EFFECT function, and compare the result with the general calculation, follow these steps:  Start Microsoft Excel and open a new worksheet.  Type the following data in cells A1:B4, and format the cells as shown:   A1: Principal       B1: $1000.00 A2: NomRate        B2:   10.00% A3: Frequency      B3:     1 A4: Term           B4:     1  Create names for the cells B1:B4 that use the labels in A1:A4. To do this, select each cell (B1:B4) separately, point to Name on the Insert menu, and then click Define.  Type the following data in cells A6 and B6: <pre class="fixed_text">  A6: Future Value    B6: =Principal*(1+(NomRate/Frequency))^(Frequency*Term) </li>  Type the following data in cells A8 and B8: <pre class="fixed_text">  A8: Future Value    B8: =Principal*(1 + EFFECT(EFFECT(NomRate,Frequency)*Term,Term)) </li>  Type the following data in cells A9 and B9: <pre class="fixed_text">  A9: Effective Rate  B9: =EFFECT(EFFECT(NomRate,Frequency)*Term,Term) </li> Format cells B6 and B8 as Currency, and cell B9 as Percentage with two decimal places.</li> Change the Term (cell B4) to 2 years to observe the effect of annual compounding.</li> Change the Frequency (cell B3) to 2 per year to observe the effect of intrayear compounding.</li></ol>

Calculation of Loan or Mortgage Payments
Loans and mortgages are usually paid down or amortized over a certain period, typically 15 to 30 years, with a fixed number of equal payments. The schedule of payments is usually presented in an amortization table, which shows the principal owing, the interest accrued, the payment required, and the balance owing after payment, for each period of the amortization. You can use the PMT worksheet function to determine the payment that is required to amortize a loan over a given period.

To construct an amortization table for a loan that uses the worksheet data above and assumes 12 equal monthly payments, follow these steps: <ol> Type the value 1 in cells B3 and B4.</li>  Type the following data in cells A5 and B5: <pre class="fixed_text">  A5: NumberOfPmts   B5: 12 </li> Select cell B5. On the Insert menu, point to Name, and then click Define. Click OK.</li>  Type the following data in cells A11 and B11 <pre class="fixed_text">  A11: Payment       B11: =PMT(NomRate/NumberOfPmts, NumberOfPmts*Term, -Principal) and notice the payment of $87.92.

NOTE: The minus (-) sign is assigned to Principal to indicate that the Payment and Principal flow in opposite directions. </li> Select cell B11. On the Insert menu, point to Name, and then click Define. Click OK.</li>  Type the following data in cells A13:D13: <pre class="fixed_text">  A13: Principal     B13: Interest     C13: Payment     D13: Balance </li>  Type the following data in cells A14 and B14: <pre class="fixed_text">  A14: =Principal    B14: =A14 * NomRate/NumberOfPmts </li>  Type the following data in cells C14 and D14: <pre class="fixed_text">  C14: =Payment      D14: =A14+B14-C14 </li> Format cells A14:D14 as Currency.</li> In cell A15, type =D14 .</li> Select cells B14:D14, click the fill handle, and fill down one row.</li> <li>Select cells A15:D15, click the fill handle, and fill down 10 rows.

Notice that the final balance is $0.00.</li> <li>Select cell C27 and double-click the AutoSum button to calculate the total amount paid in interest.</li></ol>

To see the difference between nominal and effective interest rate, suppose that no payments were actually made, and fill cells C14:C25 with zeros. When you do this, notice that the accumulated interest becomes $104.71, corresponding to an effective interest rate of 10.47% compared to the 10.00% nominal rate used in the calculation. The interest is in fact compounded monthly, so that the overall effect is one of a higher rate.

To see the correspondence between the EFFECT and NOMINAL worksheet functions, follow these steps: <ol> <li> Type the following data in cells A29 and B29: <pre class="fixed_text">  A29: Effective    B29: =EFFECT(NomRate,NumberOfPmts) </li> <li>Select cell B29. On the Insert menu, point to Name, and then click Define. Click OK.</li> <li> Type the following data in cells A30 and B30: <pre class="fixed_text">  A30: Nominal      B30: =NOMINAL(Effective,NumberOfPmts) </li> <li>Format cells A30 and B30 as Percentage with two decimal places.</li></ol>

Different Jurisdictions
The above example calculates payments by using the PMT worksheet function, as it is commonly applied in the United States. However, other jurisdictions may calculate mortgage payments differently.

For example, in Canada, the nominal interest rate would be applied semi-annually to the example above, although it would still be described as a loan at 10.00%.

In this case, you would type the value 2 in cell B3 (Frequency) to describe the two compounding periods annually in the nominal rate.

Also, the PMT function in cell B11 would have the following arguments: <pre class="fixed_text">   =PMT((1+NomRate/Frequency)^(Frequency/NumberOfPmts)-1, NumberOfPmts*Term, -Principal) And the interest calculation in cells B14 and below would have the following form: <pre class="fixed_text">   =A14 * ((1+NomRate/Frequency)^(Frequency/NumberOfPmts)-1) This difference results in slightly lower interest costs for the borrower in the Canadian situation, and illustrates the importance of knowing the financial environment in which you are using the worksheet functions.

<div class="references_section">